Attribute VB_Name = "modCloseAlert"
Private Declare Function FindWindow Lib "user32" Alias "FindWindowA" (ByVal lpClassName As String, ByVal lpWindowName As String) As Long
Private Declare Function EnumChildWindows Lib "user32" (ByVal hWndParent As Long, ByVal lpEnumFunc As Long, ByVal lParam As Long) As Long
Private Declare Function GetWindowText Lib "user32" Alias "GetWindowTextA" (ByVal hWnd As Long, ByVal lpString As String, ByVal cch As Long) As Long
Private Declare Function GetWindowTextLength Lib "user32" Alias "GetWindowTextLengthA" (ByVal hWnd As Long) As Long
Private Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hWnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Const BM_CLICK As Long = &HF5


Function EnumChildProc(ByVal hWnd As Long, ByVal lParam As Long) As Long
    Dim buf As String

    buf = String(GetWindowTextLength(hWnd) + 1, Chr$(0))
    GetWindowText hWnd, buf, Len(buf)

    If Left(buf, 9) = "Don't Fix" Then
        SendMessage hWnd, BM_CLICK, 0, 0
        EnumChildProc = 0
    Else
        EnumChildProc = 1
    End If
End Function

Public Sub CloseAlertCS2()
    Dim hWnd As Long

    hWnd = FindWindow(vbNullString, "Adobe InDesign")
    If hWnd > 0 Then
        EnumChildWindows hWnd, AddressOf EnumChildProc, ByVal 0&
    End If
End Sub

